Lezione 4 (parte terza)

Enrico Bertolazzi

```
> # integrazione metodo dei trapezi
  restart:
```

```
> # polinomio approssimante
  p := x -> a + b * x + c * x^2 ;
```

$$p := x \to a + b\,x + c\,x^2 \quad (1)$$

```
> # condizioni di interpolazione
  EQ0 := p(X[k])   = f(X[k]) ;
  EQ1 := p(X[k+1]) = f(X[k+1]) ;
  EQ2 := p(X[k+2]) = f(X[k+2]) ;
```

$$EQ0 := a + b\,X_k + c\,X_k^2 = f(X_k) \quad (2)$$

$$EQ1 := a + b\,X_{k+1} + c\,X_{k+1}^2 = f(X_{k+1})$$

$$EQ2 := a + b\,X_{k+2} + c\,X_{k+2}^2 = f(X_{k+2})$$

```
> # calcolo il polinomio interpolante
  res := solve({EQ0,EQ1,EQ2},{a,b,c}) :
  pf  := simplify(expand(subs(res,p(x)))) ;
```

$$pf := \left(X_k X_{k+2}^2 f(X_{k+1}) - X_k f(X_{k+2}) X_{k+1}^2 - X_k^2 X_{k+2} f(X_{k+1}) + X_k^2 X_{k+1} f(X_{k+2}) + f(X_k) X_{k+2} X_{k+1}^2 - f(X_k) X_{k+1} X_{k+2}^2 + x X_k^2 f(X_{k+1}) - x X_k^2 f(X_{k+2}) + x f(X_k) X_{k+2}^2 - x X_{k+2}^2 f(X_{k+1}) - x f(X_k) X_{k+1}^2 + x f(X_{k+2}) X_{k+1}^2 - x^2 X_k f(X_{k+1}) + x^2 X_{k+1} f(X_k) - x^2 X_{k+2} f(X_k) + x^2 X_{k+2} f(X_{k+1}) + x^2 X_k f(X_{k+2}) - x^2 X_{k+1} f(X_{k+2})\right) \Big/ \left(-X_k X_{k+1}^2 - X_{k+2} X_k^2 + X_{k+2} X_{k+1}^2 + X_k X_{k+2}^2 - X_{k+1} X_{k+2}^2 + X_{k+1} X_k^2\right) \quad (3)$$

```
> # se i nodi sono equispaziati si semplifica
  pf1 := simplify(expand(subs(X[k+1]=X[k]+h,X[k+2]=X[k]+2*h,pf))) ;
```

$$pf1 := -\frac{1}{2}\,\frac{1}{h^2}\left(-2 f(X_k) h^2 + 2 X_k^2 f(X_k + h) + 4 X_k f(X_k + h) h - X_k f(X_k + 2h) h + 2 x f(X_k) X_k - 4 x f(X_k + h) X_k + 2 x f(X_k + 2h) X_k - 3 f(X_k) X_k h - 4 x f(X_k + h) h + x f(X_k + 2h) h - x^2 f(X_k) + 2 x^2 f(X_k + h) - x^2 f(X_k + 2h) - X_k^2 f(X_k + 2h) - f(X_k) X_k^2 + 3 x f(X_k) h\right) \quad (4)$$

```
> # integro il polinomio interpolante
  simplify(int(pf1,x=X[k]..X[k]+2*h)) ;
```

$$\frac{1}{3} h \left(f\left(X_k\right) + 4 f\left(X_k + h\right) + f\left(X_k + 2 h\right)\right) \quad (5)$$

```
> # applicando la formula precendete sugli intervalli
  # a due a due ottengo la regola di Simpson
  simpson := proc (f,a,b,n)
    local res, h ;
    h   := (b-a)/n ;
    res := (h/3)*(f(b)+f(a)) ;
    res := res+
       (4*h/3)*add(f(a+(2*i+1)*h),i=0..(n-2)/2);
    res := res +(2*h/3)*add(f(a+(2*i)*h),i=1..(n-2)/2);
    return res ;
  end proc :
> # funzione da integrare
  f := x -> x/(1+x^2);
```

$$f := x \rightarrow \frac{x}{1 + x^2} \quad (6)$$

```
> # intervallo di integrazione
  a,b := 0, 10 ;
```

$$a, b := 0, 10 \quad (7)$$

```
> # calcolo dell'integrale esatto
  ESATTO := evalf(int(f(x),x=a..b)) ;
```

$$ESATTO := 2.307560258 \quad (8)$$

```
> # calcolo dell'integrale approssimato
  # col metodo di Simpson
  APPROSSIMATO := evalf(simpson(f,a,b,100)) ;
```

$$APPROSSIMATO := 2.307563675 \quad (9)$$

```
> evalf(ESATTO-APPROSSIMATO) ;
```

$$-0.000003417 \quad (10)$$

```
> # stima del numero di intervalli
  with(plots) :
> ddddf := (D@@4)(f) ;
```

$$ddddf := x \rightarrow -\frac{480 x^3}{\left(1 + x^2\right)^4} + \frac{120 x}{\left(1 + x^2\right)^3} + \frac{384 x^5}{\left(1 + x^2\right)^5} \quad (11)$$

```
> plot(ddddf(x),x=a..b) ;
```




```
> # costante che maggiora il modulo di ddddf in [a,b]
  M := 20 ;
```

$$M := 20 \quad \textbf{(12)}$$

```
> # con la formula dell'errore stimo gli intervalli
  N := ( (b-a)^5 * M / (180 * epsi) )^(1/4) ;
```

$$N := \frac{1}{180}\, 180^{3/4}\, 2000000^{1/4} \left(\frac{1}{epsi}\right)^{1/4} \quad \textbf{(13)}$$

```
> evalf(subs( epsi=10^(-4),N)) ;
```

$$102.6690096 \quad \textbf{(14)}$$

```
> evalf(ESATTO-simpson(f,a,b,104)) ;
```

$$-0.000002915 \quad \textbf{(15)}$$